

Modeling the quantization effects in digital filters via OrCad PSpice v. 15.7

prof. Ing. Dalibor Bielek, CSc., FVT UO Brno, K217, Kounicova 65, 612 00 Brno
Ing. Viera Biolková, FEKT VUT Brno, UREL, Purkyňova 118, 612 00 Brno
doc. Dr. Ing. Zdeněk Kolka, FEKT VUT Brno, UREL, Purkyňova 118, 612 00 Brno

INTRODUCTION

The design and subsequent analysis of digital filters is well supported by MATLAB combined with the Signal Processing Toolbox. In addition, owners of the Filter Design Toolbox can easily include quantization effects in the design. Additional features are enabled by exporting the designed realization structures into Simulink.

However, not everyone can avail themselves of such expensive software tools. That is why the paper is focused on the analysis of digital filters, including quantization effects, via the SPICE-family programs, namely via the evaluation version of OrCad PSpice from v. 15.7, which is freely available. Compared with MATLAB, another advantage is the graphical tool in the form of schematic editor, enabling easy drawing of filter realization structures with subsequent analysis of frequency responses from the input into an arbitrary node of filter structure. This feature is especially desirable for dynamic range optimization and scaling. The filter structure is created by connecting the blocks of delay, coefficient multiplication, and signal summation.

The techniques of digital filter analysis via Micro-Cap program are described in [Bio06]. These procedures were transferable to the OrCad PSpice program only for idealized analysis, i.e. without considering the quantization effects. The reason was the absence of special mathematical functions for quantization modeling in the OrCad version 10.5 and older. However, the set-up has changed beginning with version 15.7. In this paper, methods of modeling the quantization effects in OrCAD Pspice v. 15.7 are described both for the AC and the transient analyses of digital filters.

MODELING THE QUANTIZATION EFFECTS IN ORCAD PSPICE V. 15.7

Starting with version 15.7, OrCad PSpice includes the CEIL, FLOOR, and INTQ functions for behavioral modeling. The FLOOR function returns an integer value of the argument [Orc06]. It can be useful for modeling various methods of data quantization in fixed-point representation. Three of them are shown in Fig. 1 [Mit06], [Dav95].

The (a), (b), and (c) –type quantizations can be modeled in PSpice via FLOOR and other PSpice functions as follows:

$$Q(x) = \delta \text{ FLOOR}(x / \delta + 0.5) \quad (1a)$$

$$Q(x) = \delta \text{ FLOOR}(x / \delta) \quad (1b)$$

$$Q(x) = \text{SGN}(x) \delta \text{ FLOOR}(\text{ABS}(x) / \delta) \quad (1c)$$

The above methods can be used for modeling the quantization of the filter coefficients, the input data, and the arithmetic rounding errors.

A concrete PSpice implementation by the user function ROUND can be as follows. In this demonstration, rounding method (a) is used:

$$\text{.func round}(x,N) \{ \text{if}(N \leq 0, x, 2^{(-N)} * \text{FLOOR}(x * 2^N + 0.5)) \} \quad (2)$$

The number of bits N is defined by a global parameter. In Eq. (2), the user can discard the quantization after setting $N \leq 0$.




Figure 1 Quantization methods: (a) rounding, (b) two's complement truncation, (c) one's-complement and sign-magnitude truncation. Here $\delta = 2^{-N}$, N is the number of bits without the sign bit. Adopted from [Mit06].

IMPLEMENTATION OF BASIC BLOCKS IN PSPICE

Since the basic circuit variables in the SPICE-family programs are voltages and currents, the summing block, the block of coefficient multiplication, and the block of delay by sampling period $T_s = 1/f_s$ can be implemented by controlled sources. For the popular PSPICE program, the E-type controlled source can be applied:

- Summing block:

$$E_{\text{sum out 0 value}} = \{V(\text{in1}) + V(\text{in2})\}$$

- Block of multiplying by constant A :

$$E_{\text{mul out 0 in 0}} \{A\}$$

- Delay block:

$$E_{\text{z out 0 LAPLACE}} \{V(\text{in})\} \{\exp(-s/f_s)\}$$

Both constant A , representing a concrete filter coefficient, and sampling frequency f_s must be defined as global variables by *.param* command.

Before the AC analysis of a digital filter compiled from the above blocks its input must be excited by an independent voltage source with the attribute $AC = 1$. After that, one will observe the frequency dependence of voltages at the outputs of respective blocks. With regard to the principal necessity of solving the linear model, it is necessary to disable the quantization of filter variables, except for the coefficients quantization. Full quantization can be used for the Transient analysis.

It is possible to model the above blocks as SPICE subcircuits and to assign them schematic symbols in CAPTURE. We get a powerful tool for digital filter analysis.

One possible form of simple PSPICE library is given below:

```
*Digfil.lib - PSPICE library for digital filters analysis
*including the quantization effects

*quantization function, rounding
.func quant(x,N) {if(N<=0,x,2^(-N)*FLOOR(x*2^N+0.5))}
*quantization function, two's complement truncation
;.func quant(x,N) {if(N<=0,x,2^(-N)*FLOOR(x*2^N))}
```

*quantization function, ones' -complement and sign-magnitude truncation
 ;.func quant(x,N) {if(N<=0,x,SGN(x)*2^(-N)*FLOOR(ABS(x)*2^N))}

*summing block
 *this subcircuit uses global parameter Nsum
 *which MUST be defined in circuit file
 .subckt SUM in1 in2 out params:Nsum={Nsum}
 Esum out 0 value={quant(V(in1)+V(in2),Nsum)}
 .ends

* multiplier
 *this subcircuit uses global parameters Ncoef and Nmul
 *which MUST be defined in circuit file
 .subckt MUL in out params:coef=1 Ncoef={Ncoef} Nmul={Nmul}
 Emul out 0 value={quant(V(in)*quant(coef,Ncoef),Nmul)}
 .ends

*delay block
 *this subcircuit uses global parameter fs
 *which MUST be defined in circuit file
 .subckt DELAY in out
 Ez out 0 LAPLACE {V(in)} {exp(-s/fs)}
 Raux out 0 1T
 .ends

DEMONSTRATION OF PSPICE SIMULATION




Figure 2 IIR filter of LP type [Bio06].

The following demonstration will be performed on the digital filter in Fig. 2 [Bio06]. It is a cascade IIR 8th - order low-pass filter with a cut-off frequency of 3.4kHz and with a sampling frequency of 22.05kHz. The filter coefficients are given in the PSpice input file:

```

PSpice circuit file: IIR 8-th order low-pass digital filter, full quantization
*
.options ITL4 100
.param fs 22.05kHz N 7 Ncoef 0 Nsum {N} Nmul {N}
.param a11 1.8880173933198123 a12 0.99999999999998335
+   b11 -0.32655932340433669 b12 0.039446152567426181
+   a21 1.2426043302792451 a22 1.00000000000000131
+   b21 -0.4147385903741958 b22 0.15472564825866436
+   a31 0.62610036993806006 a32 0.99999999999999201
+   b31 -0.57140840958280426 b32 0.38161259426751648
+   a41 0.3148879700393597 a42 0.99999999999999423
+   b41 -0.77820865449734911 b42 0.74175331275429501
.param gain 0.0053733804257323015

;Vin in 0 DC 1 AC 1 ;input signal for step response and AC analysis
Ein in 0 value={stp(time)-stp(time-1/fs)};input signal for impulse response
Xgain in in1 MUL params:coef={gain}
*stage No. 1
Xz11 11 12 DELAY
Xz12 12 13 DELAY
Xb11 12 m11 MUL params:coef={-b11} ;the feedback path
Xb12 13 m12 MUL params:coef={-b12}
Xsum11 in1 s11 11 SUM
Xsum13 m11 m12 s11 SUM
Xa11 12 m13 MUL params:coef={a11} ;the forward path
Xa12 13 m14 MUL params:coef={a12}
Xsum12 11 s12 out1 SUM
Xsum14 m13 m14 s12 SUM
*end of stage No. 1
*stage No. 2
Xz21 21 22 DELAY
Xz22 22 23 DELAY
Xb21 22 m21 MUL params:coef={-b21} ;the feedback path
Xb22 23 m22 MUL params:coef={-b22}
Xsum21 out1 s21 21 SUM
Xsum23 m21 m22 s21 SUM
Xa21 22 m23 MUL params:coef={a21} ;the forward path
Xa22 23 m24 MUL params:coef={a22}
Xsum22 21 s22 out2 SUM
Xsum24 m23 m24 s22 SUM
*end of stage No. 2
*stage No. 3
Xz31 31 32 DELAY
Xz32 32 33 DELAY
Xb31 32 m31 MUL params:coef={-b31} ;the feedback path
Xb32 33 m32 MUL params:coef={-b32}
Xsum31 out2 s31 31 SUM
Xsum33 m31 m32 s31 SUM
Xa31 32 m33 mul params:coef={a31} ;the forward path
Xa32 33 m34 mul params:coef={a32}
Xsum32 31 s32 out3 SUM

```

```

Xsum34 m33 m34 s32 SUM
*end of stage No. 3
*stage No. 4
Xz41 41 42 DELAY
Xz42 42 43 DELAY
Xb41 42 m41 MUL params:coef={-b41} ;the feedback path
Xb42 43 m42 MUL params:coef={-b42}
Xsum41 out3 s41 41 SUM
Xsum43 m41 m42 s41 SUM
Xa41 42 m43 MUL params:coef={a41} ;the forward path
Xa42 43 m44 MUL params:coef={a42}
Xsum42 41 s42 out4 SUM
Xsum44 m43 m44 s42 SUM
*end of stage No. 4

;step param Ncoef list 0 7 8 9
;.AC dec 1000 10 11.025k
.tran 0 5m skipbp
.probe
.inc digfil.lib

.end

```

The results of AC analysis are in Fig. 3. Note that it is necessary to define filter coefficients by at least 9 bits in order to achieve the required characteristic within the entire frequency range.




Figure 3 Frequency responses of filter in Fig. 2 for several numbers of bits for coefficient quantization.

The filter impulse response as a result of Transient analysis is in Fig. 4. The filter coefficients are quantized to 9 bits. A comparison of two cases is given, for rounding free arithmetic operation (case (a)), and for 7-bit quantization of the data (case (b)). For the case (b), one can observe the parasitic limit cycle operation. Note that the waveforms represent

the continuous-time representation of discrete-time signal. That is why one must allow for the steady-state values within a concrete sampling action.




Figure 4 Impulse responses of filter in Fig. 2 for Ncoef=9 and (a) rounding free arithmetic operation, (b) for 7 bits quantization.

CONCLUSION

The above facilities of digital filter analyses via the SPICE-family programs have been verified for many simulation problems. Additional details, e.g. how to simulate the quantization noise propagation from the digital filter input to its output, etc., are described in [Bio06].

Acknowledgment

This work is supported by the Grant Agency of the Czech Republic under grants No. 102/05/0771 and No. 102/05/0277, and by the research programmes of BUT MSM0021630503, MSM0021630513, and UD Brno MO FVT0000403.

References

- [Bio06] BIOLEK, Dalibor, ABUETWIRAT, Ines Faisal. Analysis of digital filters via SPICE-compatible programs, 2006. Elektrorevue, 2006/26, ISSN 1213-1539, <http://www.elektrorevue.cz/clanky/06026/english.htm>.
- [Orc06] Pspice A/D Reference Guide. Product Version 15.7, July 2006.
- [Mit06] MITRA, Sanjit K. Digital Signal Processing. A Computer-Based Approach. Third Edition. McGraw Hill, 2006. ISBN 007-124467-0.
- [Dav95] DAVÍDEK, Vratislav, PŠENIČKA, Bohumil. Tools for discrete-time signal processing. Exercises. ČVUT Praha, 1994. ISBN 80-01-01139-9 (In Czech).